Tutorial 10: Using Data Stored in ROM

New Instructions Introduced

MACP

MVPD

READA

Overview of Tutorial

So far we have worked exclusively on data stored in data memory (RAM). However, data memory is a limited resource. Certain variables need only to be read and not written. These can be stored in program memory (ROM) rather than in data memory.

In this tutorial we shall see how to copy variables from program memory to data memory and how to work directly on variables stored in program memory.

Copying a Block of Data from ROM to RAM using Direct Addressing

One way to copy a word of data from a program memory address to a data memory address is to use the instruction MVPD (move data from program memory to data memory). This can be used with direct addressing as shown in Example 10-1:

Example 10-1.

	
	

	LD #3, DP
	; Page 3. Gain access to data memory

; addresses 180h to 1FFh.

	MVPD 1000h, 8h
	; Move word of data from program memory

; address 1000h to the data

; memory address 180h + 8h = 188h.


When using the instruction MVPD (move data from program memory to data memory) we need to specify two addresses as operands: the source address in program memory we wish to copy from and the destination address in data memory we wish to copy to. 

Just as the term data memory address is abbreviated to dmad in the Texas Instruments databooks, the term program memory address is often abbreviated to pmad.

Using direct addressing in order to copy two sequential words of data from program memory to data memory we would repeat the instruction MVPD, as shown in Example 10-2:

Example 10-2.

	
	

	LD #3, DP
	; Page 3. Gain access to data memory

; addresses 180h to 1FFh. 

	MVPD 1000h, 10h
	; Move the word of data from program

; memory address 1000h to the data

; memory address 180h + 10h = 190h.

	MVPD 1001h, 11h
	; Move the word of data from program

; memory address 1001h to the data

; memory address 180h + 11h = 191h.


A line of code is required for each value to be copied. In order to copy a large number of words of data, we would require a large number of instructions. This is both laborious and wasteful of code. A more efficient way is to use indirect addressing.

Copying a Block of Data from ROM to RAM using Indirect Addressing

The instruction MVPD (move data from program memory to data memory) also supports indirect addressing. Example 10-3 shows how to copy a single word from a program memory address to a data memory address:

Example 10-3.

	
	

	STM #185h, AR6
	; Store the address of data memory

; location in auxiliary register AR6.

; AR6 = 185h. 

	MVPD 1000h, *AR6
	; Copy the word at program memory

; address 1000h to the data memory

; address pointed to by AR6. Data memory

; address 185h now contains the same

; value as that contained at program

; memory address 1000h.


In order to efficiently copy a block of data from program memory to data memory, we can use the instruction RPT (repeat). In this case we shall copy 10 words of data from the program memory address 1000h to the block of data memory starting at the address 200h.

Example 10-4. 

	
	

	STM #200h, AR6
	; Store address of start of data

; memory block in auxiliary

; register AR6. AR6 = 200h. 

	RPT #(10-1)
	; Execute next instruction and then 

; repeat nine times.

	MVPD 1000h, *AR6+
	; Starting at program memory 

; address 1000h, copy 10 successive

; words to the block of data

; memory starting at the data memory

; address 200h. The last word is

; written to the data memory address

; 209h.


We place the instruction RPT before the instruction MVPD. The operand used with the instruction RPT needs to be one less than the total number of words to be copied. In Example 10-4, ten data words are to be copied from program memory to data memory, so 10-1 = 9 repeats are required.  

The instruction MVPD has its own program address register (PAR). When the instruction MVPD is executed, the program memory address supplied as the operand is passed to the program address register. The PAR is increased automatically once every repeat. On the first repeat, the contents of program memory address 1001h will be copied. On the second repeat, a word will be copied from program memory  address 1002h. This means a word is copied from a different program memory address every time. The final word will be copied from program memory address 1009h.  

We use the operand *AR6+ to increment the value of the pointer AR6 every time we execute the instruction MVPD. Each successive word from program memory will be stored at the next higher data memory address. 

Copying a block of data using the instruction RPT (repeat) does not function correctly when used with direct addressing. The fragment of code in Example 10-5 is intended to use direct addressing to copy 10 words from the program memory address 1000h to the data memory addresses starting at 200h:

Example 10-5.

	
	

	LD #4, DP
	; Use data page 4. Gain access to data

; memory addresses 200h to 27Fh. 

	RPT #(10-1)
	; Execute the next instruction ten

; times.

	MVPD 1000h, 0h
	; Starting at program memory address

; 1000h, attempt to copy 10

; successive words to the data memory

; addresses starting at 200h. The last

; word is in fact written to the data

; memory address 200h. When used with

; direct addressing, there is no

; automatic increment of the program

; memory address.


Unfortunately, this fragment of code will always write to the data memory address 200h. Direct addressing does not provide a mechanism to increment the data memory address. Every value will be written to data memory address 200h. At the end of the repeat sequence, data memory address 200h will contain the 10th value from the table in program memory. 

Hence, when moving multiple program words from program memory to data memory using the instruction MVPD, indirect addressing should be used.

Using Look-up Tables

Consider the case of a system where the output is derived mathematically from the input. If there is no simple relationship between the input and the output, this makes the calculation of the output difficult. However, it may be possible to describe a non-linear relationship between an input and an output in the form of a look-up table. This has the advantage over a complex calculation in that it uses few instructions and is therefore quick to execute. In DSP applications, look-up tables are widely used to generate sine / cosine functions, where the calculation would be much slower and more wasteful of code. 

One way we can implement a look-up table in C is to put the table into an array. In order to look-up an element from the table, we read the specific element using an index as shown in Example 10-6:

Example 10-6.

	
	

	unsigned int table[12]={10,12,15,18,
	// Array of 12

	                        22,27,33,39,
	// elements with  

	                        47,56,68,82};
	// constants. 

	unsigned int index;
	

	unsigned int output;
	

	index = 7;
	// Work on 8th

// element

	output = table[index];
	// Read from table

// using index.


In the case of a look-up table containing 12 elements, the elements are numbered from 0 to 11. To address a particular element, we put the number of the element in the variable index. The first element of the array is to be found in table[0] and the last element in table[11].  

We shall now implement Example 10-6 using TMS320C5000 assembly language. We shall adhere to the C convention for our look-up table by storing the first value of the table at the starting address of the table.

One way to read a value from the table is to use the instruction MVPD. This instruction is really designed to copy blocks of data. The problem here is that the exact address of the table entry must be known and cannot be calculated in the course of the program. 

To extract a single value from a look-up table in program memory, a better way is to use the instruction READA (read program memory addressed by accumulator A and store in data memory). The instruction READA takes a single operand which is the destination address in data memory where the value is to be written. We cannot use accumulator B.

The assembly language implementation of Example 10-6 is given in Example 10-7: 

Example 10-7.

	
	

	table:
	.WORD 10      ; element 0            

	 
	.WORD 12      ; element 1

	
	.WORD 15      ; element 2

	
	.WORD 18      ; element 3  

	
	.WORD 22      ; element 4  

	
	.WORD 27      ; element 5 

	
	.WORD 33      ; element 6   

	
	.WORD 39      ; element 7 

	
	.WORD 47      ; element 8 

	
	.WORD 56      ; element 9 

	
	.WORD 68      ; element 10

	
	.WORD 82      ; element 11

	lookup:
	

	LD #8, DP
	; Page 8. Gain access to data memory

; addresses 400h to 47Fh.

	ST #7, 20h
	; Store a number in the range 0 to 11 in

; the variable index. We shall use 7. 

	LD 20h, A
	; Our index is in data memory address

; 400h + 20h = 420h. Copy to accumulator

; A.

	ADD #table, A
	; Accumulator A contains the value at the

; address table + index, in this case

; table + 7. The value 39 is stored at

; this address.  

	READA 21h
	; Read the value from table corresponding

; to index and store at data memory 

; address 421h. Data memory address 421h

; now contains 39. 


The constants for our look-up table are stored in program memory by the using the .WORD assembler directive. In order to read a constant from the look-up table, we load accumulator A with the starting address of the table then add the index.  

For example, to read the constant 27 we would use an index of 5.

We could also write the portion of the code in Example 10-7 that extracts a data word from the lookup table as follows:

Example 10-8.

	
	

	LD #table, A
	; Load accumulator A with the address of 

; the start of the table.

	ADD 20h, A
	; Add index in data memory address 421h

; to the starting address of table.

; Accumulator A contains address

; of table + index. 

	READA 21h
	; Read value from table corresponding to

; index and store at data memory address

; 421h. 


The only difference between Examples 10-7 and 10-8 are the order in which accumulator A has been loaded.

Care does need to be taken when using look-up tables. There is no range check for the index, so if this were to contain an out-of-range value, then the instruction READA would fetch a value from outside of the look-up table. A value from program memory not intended to be part of the look-up table would be used. 

Keeping Look-up Tables Separate from Code

When the assembler reaches a look-up table in program memory, it has no way of telling whether the code contains a look-up table or simply instructions to be executed. Consider the fragment of code shown in Example 10-9 which contains a table of four constants:

Example 10-9.

	
	

	table1:
	.WORD 0

	
	.WORD 1

	
	.WORD 2

	
	.WORD 3

	lookup:
	


The code in Example 10-9 can in fact be interpreted by the assembler as:

Example 10-10.

	
	

	table1:
	ADD 0h, A

	
	ADD 1h, A

	
	ADD 2h, A

	
	ADD 3h, A

	lookup:
	


Caution should therefore be taken when reading a value from a look-up table. If the index used is greater than the number of entries in the table, then executable code will be read rather than constants. Should the number of entries in a look-up table be reduced, then the maximum value of the index should also be reduced.

Normally, look-up tables would be kept separate from executable code, for example at the end of the program. However, it may be necessary to place the look-up table within a block of executable code. In  this case, the instruction B (branch) should also be used: 

Example 10-11.

	
	
	

	
	B lookup2
	; Skip the code in table2 by 

; branching to the label

; lookup2. 

	table2:
	.WORD 0h
	; These four words are 

	
	.WORD 1h
	; only accessible using

	
	.WORD 2h
	; the instructions MVPD 

	
	.WORD 3h
	; or READA.

	lookup2:
	
	


Using the instruction B (branch), the look-up table table2 is skipped, except when it is being read as a table. 

Multiplication with Accumulation using Data in Program Memory 

In the previous tutorial we saw how to multiply and accumulate. We always multiplied a value in data memory with another value also in data memory. However in a program, numbers such as filter constants would normally be stored in program memory. We could copy them to data memory and then manipulate them, but it is often more convenient to work on them directly. 

The instruction MACP (multiply by program memory and accumulate) allows us to multiply a value stored in data memory by a variable stored in program memory, then accumulate the product in one of the accumulators. Example 10-12 illustrates how this is done.  

Example 10-12.

	
	

	LD #2, DP
	; Page 2. Access to data memory

; addresses 100h to 17Fh.

	MACP 25h, table1, A
	; Multiply the value at data

; memory address 125h by the

; value contained at program

; memory address table1. Add

; the product to accumulator A.   


The instruction MACP takes three operands. The first operand is the address in data memory and the second operand is the address in program memory. The third operand is the destination and can be either accumulator A or accumulator B. 

We can also use indirect addressing, this time with accumulator B as the destination:

Example 10-13.

	
	

	STM #125h, AR3
	; Store address of data memory

; location in auxiliary register

; AR3. AR3 = 125h.

	MACP *AR3, table1, B
	; Multiply the value at the data

; memory address pointed to 

; by AR3 by the contents of the

; program memory address table1.

; Add the product to accumulator

; B. Automatically point to the

; next address in table1.   


Like the instruction MVPD, the instruction MACP also uses the special register known as the Program Address Register (PAR). This is loaded with the second operand. When used with the instruction RPT (repeat), every time the instruction MACP is executed, the value in PAR is incremented. This means that the address in program memory is automatically updated. To copy a block, all we need to do is to specify the starting address of the table in program memory. The address in program memory is automatically updated for us by the instruction RPT.

To multiply and accumulate 16 words of a data block beginning at data memory address 280h with a set of constants starting at program memory address table1 we write:  

Example 10-14.

	
	

	STM #280h, AR4
	; Store the address of the start

; of the data memory block in

; auxiliary register AR4.

; AR4 = 280h. 

	LD #0, A
	; Start accumulation with zero in

; accumulator A. 

	RPT #(16-1)
	; Execute following operation and 

; repeat 15 times.

	MACP *AR4+, table1, A
	; Multiply the value at the data

; memory address pointed to

; by AR4 by the value at program

; memory address table1. Add the

; product to accumulator A.

; Increment the auxiliary register

; AR4 to point to the next address

; in data memory. The program 

; memory address is incremented 

; automatically.   


When we carry out the first multiplication with accumulation, we add the product to whatever value is in accumulator A. We therefore start by clearing accumulator A to zero so that our accumulation begins at zero. We use the expression *AR4+ to indicate that auxiliary register AR4 is used as a pointer to a block of data memory and the address is incremented after each read. Each accumulation will therefore work on a different data memory address.

We could also carry out the same operation as shown in Example 10-14, this time using accumulator B. This time we have combined the instructions LD #0, A and RPT #(16-1) into the single instruction RPTZ B, #(16-1).
Example 10-15.

	
	

	STM #280h, AR4
	; Store address of start of block

; of data memory in auxiliary

; register AR4. AR4 = 280h. 

	RPTZ B, #(16-1)
	; Clear accumulator B then execute

; the following operation 16

; times.

	MACP *AR4+, table1, B
	; Multiply the value at the data

; memory address pointed to

; by AR4 by the contents of

; program memory address table1.

; Add the product to accumulator

; B. Increment auxiliary register

; AR4 to point to the next address

; in data memory. The program

; memory address is incremented

; automatically.   


Example 10-15 shows how well the TMS320C5000 architecture is suited to multiplications with accumulation using values from a table in program memory.

Upgrading from the TMS320C2000 to the TMS320C5000

The syntax of instructions to work directly on values stored in program memory does vary between the two families:

Table 10-1. Comparison of Instructions

	Description
	TMS320C2000 Instruction
	TMS320C5000 

Instruction

	Move data from program memory to data memory
	BLPD
	MVPD

	Read program memory addressed by accumulator A and store in data memory
	TBLR
	READA

	Multiply accumulate
	MAC
	MACP 


The TMS320C2000 devices abbreviate the term program memory address to pma. On the other hand, the TMS320C5000 databook uses the abbreviation pmad.

The operation of multiplying a block of data memory by a set of constants in program memory is simpler on the TMS320C5000 than on the TMS320C2000. There is no need to relocate a block of data in program memory to data memory.

Questions

	1.
	Why would we wish to store constants in program memory rather than in data memory?

	2.
	What does the abbreviation pmad stand for?

a) 
positive mode address

b) 
previous memory accumulate

c) 
program memory address

	3.
	Why do we use indirect addressing rather than direct addressing when moving multiple words of data from program memory to data memory using the instruction MVPD? 

	4.
	How do we copy a word of data from program memory address 400h to data memory address 100h?

	5.
	Why would we use a look-up table?

	6.
	Both the instructions MVPD and READA move data from program memory to data memory. How does the usage of the two instructions differ?

	7.
	How do we prevent the values in a look-up table being executed as code?

	8.
	How do we increment the address of the program memory when using the instruction MACP?

	9.
	The instruction RPTZ (repeat next instruction and clear accumulator means:

a) 
Clear the accumulator once then repeat next instruction

b) 
Repeat next instruction, every time clearing the accumulator?

	10.
	Why do we use the instruction RPTZ rather than RPT before a series of 

multiplications with accumulate MACP?

